

# DISPLAY

```
DISPLAY [(rep)] [options] { [/...] [output-format] output-element}...
```

Operand	Possible Structure	Possible Formats	Referencing Permitted	Dynamic Definition
Operand1	S A G N	A N P I F B D T L G O	yes	no

Related Statements: WRITE | WRITE TITLE | WRITE TRAILER

## Function

The DISPLAY statement is used to specify the fields to be output on a report in column format. A column is created for each field and a field header is placed over the column.

## Report Specification - rep

The notation (*rep*) may be used to specify the identification of the report for which the DISPLAY statement is applicable. A value in the range 0 - 31 or a logical name which has been assigned using the DEFINE PRINTER statement may be specified. If (*rep*) is not specified, the DISPLAY statement will apply to the first report (report 0).

## options

```
[NOTITLE] [NOHDR] [ [AND] [GIVE] [SYSTEM] FUNCTIONS ] [(statement-parameters)]
```

## Page Title/NOTITLE

By default, Natural generates a single title line for each page resulting from a DISPLAY statement. This title contains the page number, the time of day, and the date. Time of day is set at the beginning of the program execution (TP mode) or at the beginning of the job (batch mode).

The default title line may be overridden by using a WRITE TITLE statement, or it may be suppressed by specifying the keyword NOTITLE in the DISPLAY statement.

Default title will be produced:

```
DISPLAY NAME
```

User title will be produced:

```
DISPLAY NAME  
WRITE TITLE 'USER TITLE'
```

No title will be produced:

```
DISPLAY NOTITLE NAME
```

If the NOTITLE option is used, it applies to all DISPLAY, PRINT and WRITE statements within the same object which write data to the same report.

## Column Headers/NOHDR

Column headers are produced for each field specified in the DISPLAY statement using the following rules:

The header text may be explicitly specified in the DISPLAY statement before the field name. For example:

```
DISPLAY 'EMPLOYEE' NAME 'SALARY' SALARY
```

If you do not specify an explicit header for a field, the header as defined in the DEFINE DATA statement will be used. If for a database field no header is defined in the DEFINE DATA statement, the default header as defined in the DDM will be used; if no default header is defined in the DDM, the field name will be used as header. If for a user-defined variable no header is defined in the DEFINE DATA statement, the variable name will be used as header. See also the DEFINE DATA statement for header definition.

```
DISPLAY NAME SALARY #NEW-SALARY
```

Natural always underlines column headings and generates one blank line between the underlining and the data being displayed.

If there are multiple DISPLAY statements in a program, the first DISPLAY statement determines the column header(s) to be used; this is evaluated at compilation time .

## Suppressing Column Headers

To suppress the column header for a single field, specify the characters '/' (apostrophe-slash-apostrophe) before the field name. For example:

```
DISPLAY '/' NAME 'SALARY' SALARY
```

To suppress all column headers, specify the keyword NOHDR:

```
DISPLAY NOHDR NAME SALARY
```

NOHDR only takes effect for the first DISPLAY, as subsequent DISPLAY statements cannot create column headers anyhow.

If both NOTITLE and NOHDR are used, they must be specified in the following order:

```
DISPLAY NOTITLE NOHDR NAME SALARY
```

## GIVE SYSTEM FUNCTIONS

The GIVE SYSTEM FUNCTIONS clause is used to make available the Natural system functions AVER, COUNT, MAX, MIN, NAVER, NCOUNT, NMIN, SUM, TOTAL. These are evaluated when the DISPLAY statement containing the GIVE SYSTEM FUNCTIONS clause is executed.

These functions may then be referred to in a statement executed as a result of an end-of-page condition.

Only one DISPLAY statement per report may contain a GIVE SYSTEM FUNCTIONS clause. When system functions are evaluated from a DISPLAY statement, they are evaluated on a page basis, which means that all functions (except TOTAL) are reset to zero when a new page is initiated.

When system functions are used within a DISPLAY statement within a subroutine, the end-of-page processing must occur within the same routine.

### statement-parameters

One or more parameters, enclosed within parentheses, may be specified.

Each parameter specified will override any previous parameter specified in a GLOBALS command, SET GLOBALS or FORMAT statement. If more than one parameter is specified, they must be separated by one or more blanks from one another. Each parameter specification must not be split between two statement lines.

The parameter settings applied here will only be regarded for variable fields, but they have no effect on text-constants. If you would like to set field attributes for a text-constant, they have to be set explicitly for this element.

#### Example:

```

DEFINE DATA LOCAL
1 VARI (A4)      INIT <'1234'>          /*      Output
END-DEFINE                                           /*      Produced
*                                                    /*      -----
DISPLAY NOHDR   'Text'                  '='  VARI          /*      Text 1234
DISPLAY NOHDR (PM=I) 'Text'              '='  VARI          /*      Text 4321
DISPLAY NOHDR   'Text' (PM=I)           '='  VARI (PM=I) /*      txeT 4321
DISPLAY NOHDR   'Text' (PM=I)           '='  VARI          /*      txeT 1234
END

```


For information on which parameters you may specify and a description of each parameter, see the section Session Parameters in the Natural Reference documentation.

## Line Advance - Slash

When specified within a text element, a slash "/" causes a line advance for the text displayed.

When specified between output elements, it causes the output element specified by "/" to be placed vertically within the same column. The header for this column will be constructed by placing the headers of the vertically displayed elements vertically above the column.

### output-format



## Field Positioning Notations


<b><i>nX</i></b>	<p>Example:          DISPLAY NAME <b>5X</b> SALARY  <b>Note:</b> (for Mainframes Only)          This notation inserts <i>n</i> spaces between columns. <i>n</i> must not be "0".</p>
<b><i>nT</i></b>	<p>This notation causes positioning (tabulation) to display position <i>n</i>. Backward positioning is not permitted.          In the following example, NAME is displayed beginning in position 25, and SALARY beginning in position 50:          DISPLAY <b>25T</b> NAME <b>50T</b> SALARY</p>
<b><i>x/y</i></b>	<p>This notation causes the next element to be placed <i>x</i> lines below the output of the last statement, beginning in column <i>y</i>.  <i>y</i> must not be "0". Backward positioning is not permitted.</p>
<b>T*<i>field-name</i></b>	<p>This notation is used to position to a specific print position of a <i>field</i> used in a previous DISPLAY statement. Backward positioning is not permitted.</p>
<b>P*<i>field-name</i></b>	<p>This notation is used to position to a specific print position <i>and line</i> of a <i>field</i> used in a previous DISPLAY statement. It is most often used in conjunction with vertical display mode. Backward positioning is not permitted.</p>

## Override Column Heading Assignment

<b>'='</b>	<p>If placed immediately before the field, '=' indicates that either the default header specified for the field in the DDM is to be used as header, or, if no default header is specified, the field name is to be used as header.</p>
<b>'text'</b>	<p>If placed immediately before a field, text overrides the column heading. The character '/' before a field causes the header for the field to be suppressed.</p> <p>DISPLAY <b>'EMPLOYEE'</b> NAME <b>'MARITAL/STATUS'</b> MAR-STAT</p> <p>If multiple text elements are specified before a field name, the last text element will be used as the column header and the other text elements will be placed before the value of the field within the column.</p>
<b>'c'(n)</b>	<p>The character <i>c</i> is displayed <i>n</i> times immediately before the field value.</p> <p>DISPLAY <b>** (5) '='</b> NAME</p>

**attributes**

Indicates the display and color attributes to be used for text display. Attributes may be:



1. Display attributes (see the session parameter AD in the Natural Reference documentation).
2. Color attributes (see the session parameter CD in the Natural Reference documentation).

## Vertical/Horizontal Display

The VERT clause may be used to cause multiple field values to be positioned underneath one another in the same column. In vertical mode, a new column may be initiated by specifying the keyword VERT or HORIZ.

The column heading in vertical mode is controlled using the entry or entries specified with the AS clause as described below.

- No column heading is produced if the AS clause is omitted.  
DISPLAY VERT NAME SALARY
- If AS 'text' is specified, text is used as the column heading. The character "/" in the character string of text will cause multiple lines of column headings.  
DISPLAY VERT AS 'LAST/NAME' NAME
- If AS 'text' CAPTIONED is specified, text is used as the column heading and the standard heading text or field name is inserted immediately before the field value in each detail display line.  
DISPLAY VERT AS 'PERSONS/SELECTED' CAPTIONED NAME FIRST-NAME
- If AS CAPTIONED is specified, the standard heading text for the field (either heading text or the field name) will be used as the column heading.  
DISPLAY VERT AS CAPTIONED NAME FIRST-NAME

Vertical and horizontal column orientation may be intermixed by using the respective keyword.

To suspend vertical display for a single output element, you may place a dash "-" in front of the element. For example:

```
DISPLAY VERT NAME - FIRST-NAME SALARY
```

In the above example, FIRST-NAME will be output horizontally next to NAME, while SALARY will be output vertically again, i.e. below NAME.

The standard display mode is horizontal. A column is constructed for each field to be displayed.

Column headings are obtained and used by Natural according to the following priority:

1. heading 'text' supplied in the DISPLAY statement;
2. the default heading defined in the DDM (database fields), or the name of a user-defined variable;
3. the field name as defined in the DDM  
(if no heading text was defined for the database field).

For group names, a group heading is produced for the entire group. When specifying a group, only the heading for the entire group may be overridden by a user-specified heading.

The maximum number of column header lines is 15.

Line size overflow is not permitted for output resulting from a DISPLAY statement. If a line overflow occurs, an error message is issued.

## output-element

$$\left[ \begin{array}{l} \{ \text{'text'} [(attributes)] \} \\ \{ \text{'c'} (n) [(attributes)] \} \dots \\ n\mathbf{X} \\ n\mathbf{T} \\ x/y \end{array} \right] [ '=' ] \{ operand1 [(parameters)] \}$$

Operand	Possible Structure	Possible Formats	Referencing Permitted	Dynamic Definition
Operand1	S A G N	A N P I F B D T L G O	yes	no

<i>nX</i>	This is the same as under <i>output-format</i> (see above).
<i>nT</i>	This is the same as under <i>output-format</i> (see above).
<i>x/y</i>	This is the same as under <i>output-format</i> (see above).
' <i>text</i> '	This is the same as under <i>output-format</i> (see above).
' <i>c</i> ' ( <i>n</i> )	This is the same as under <i>output-format</i> (see above).
' <i>text</i> ' '='	If ' <i>text</i> ' '=' is placed immediately before the field, <i>text</i> is output immediately before the field value. DISPLAY '*****' '=' NAME
<i>attributes</i>	This is the same as under <i>output-format</i> (see above).
<i>operand1</i>	The field to be displayed.  <i>Note for DLI databases:</i> <i>The DLI AIX fields can be displayed only if a PCB is used with the AIX specified in the parameter PROCSEQ. If not, an error message is returned by Natural at runtime.</i>
<i>parameters</i>	One or more parameters, enclosed within parentheses, may be specified immediately after <i>operand1</i> . Each parameter specified in this manner will override any previous parameter specified in a GLOBALS command, SET GLOBALS or FORMAT statement. If more than one parameter is specified, one or more blanks must be placed between each entry. An entry must not be split between two statement lines.

## Defaults

The following defaults are applicable for a DISPLAY statement:

1. The width of the report defaults to the value set when Natural is installed. This default value is normally 132 in batch mode or the line length of the terminal in TP mode. It may be overridden with the session parameter LS. In TP mode, line size (LS) and page size (PS) parameters are set by Natural based on the physical characteristics of the terminal type in use.
2. When the DISPLAY output is displayed on a terminal screen, the output begins in physical column 2 (because column 1 must be reserved for possible use as an attribute position on a 3270-type terminal). When the DISPLAY output is printed on paper, the printout begins in the leftmost column (column 1).
3. The default spacing factor between elements is one position. There is a minimum of one space between columns (reserved for terminal attributes). This default may be overridden with the session parameter SF.
4. The length of the field or the field heading, whichever is greater, determines the column width for the report (unless the HW parameter is used). If the field is longer than the heading, the heading will be centered over the column unless the HC=L or HC=R parameter is used to produce a left-justified or right-justified heading. If the heading is longer than the field, the field will be left-justified under the heading. The values contained in the field are left-justified for alphanumeric fields and right-justified for numeric fields. Numeric fields may be displayed left-justified by specifying AD=L. Alphanumeric fields may be displayed right-justified by specifying AD=R. In a vertical display, the longest data value or heading among all fields determines the column width (unless the HW parameter is used).

5. One extra high-order print position is reserved for a sign when printing a numeric field. The session parameter SG may be used to suppress the sign position.
6. Page overflow is checked before execution of a DISPLAY statement. No new page title or trailer information is generated during the execution of a DISPLAY statement.

## Example 1

```

/* EXAMPLE 'DISEX1:' DISPLAY (USING NX, NT NOTATION)
/*****
LIMIT 4
READ EMPLOYEES BY NAME
  DISPLAY NOTITLE 5X NAME 50T JOB-TITLE
/*****
END
    
```

NAME	CURRENT POSITION
-----	-----
ABELLAN	MAQUINISTA
ACHIESON	DATA BASE ADMINISTRATOR
ADAM	CHEF DE SERVICE
ADKINSON	SALES PERSON

## Example 2

```

/* EXAMPLE 'DISEX2' DISPLAY (GIVE SYSTEM FUNCTIONS)
/*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 NAME
  2 FIRST-NAME
  2 SALARY (1)
  2 CURR-CODE (1)
END-DEFINE
/*****
LIMIT 15
FORMAT PS=15
READ EMPLOY-VIEW
  DISPLAY GIVE SYSTEM FUNCTIONS
    PERSONNEL-ID NAME FIRST-NAME SALARY (1) CURR-CODE (1)
  AT END OF PAGE
    WRITE / 'SALARY STATISTICS:'
      / 7X 'MAXIMUM:' MAX(SALARY(1)) CURR-CODE (1)
      / 7X 'MINIMUM:' MIN(SALARY(1)) CURR-CODE (1)
      / 7X 'AVERAGE:' AVER(SALARY(1)) CURR-CODE (1)
  END-ENDPAGE
END-READ
/*****
END
    
```

PERSONNEL ID	NAME	FIRST-NAME	ANNUAL SALARY	CURRENCY CODE
50005600	MORENO	HUMBERTO	165810	FRA
50005500	BLOND	ALEXANDRE	172000	FRA
50005300	MAIZIERE	ELISABETH	166900	FRA
50004900	CAUDAL	ALBERT	167350	FRA
50004600	VERDIE	BERNARD	170100	FRA
50004300	GUERIN	MICHELE	163900	FRA
50004200	VAUZELLE	BERNARD	159790	FRA
50004100	CHAPUIS	ROBERT	169900	FRA
50004000	MONTASSIER	JEAN	175550	FRA
SALARY STATISTICS:				
	MAXIMUM:	175550	FRA	
	MINIMUM:	159790	FRA	
	AVERAGE:	167922	FRA	

### Example 3

```

* EXAMPLE 'DISEX3': DISPLAY (USING P* NOTATION)
*****
DEFINE DATA LOCAL
  1 EMP-VIEW VIEW OF EMPLOYEES
    2 NAME
    2 SALARY (1)
    2 BIRTH
    2 CITY
END-DEFINE
*
LIMIT 2
READ EMP-VIEW BY CITY FROM 'N'
  DISPLAY NOTITLE NAME CITY
    VERT AS 'BIRTH/SALARY' BIRTH (EM=YYYY-MM-DD) SALARY (1)
  SKIP 1
  AT BREAK OF CITY
    DISPLAY P*SALARY (1) AVER(SALARY (1))
  SKIP 1
END-BREAK
END-READ
END
    
```

NAME	CITY	BIRTH	SALARY
-----	-----	-----	-----
WILCOX	NASHVILLE	1970-01-01	38000
MORRISON	NASHVILLE	1949-07-10	36000
			37000

## Example 4

```

/* EXAMPLE 'DISEX4:' DISPLAY (USING 'TEXT', 'C(N)' NOTATION, AND
/*
/*                               ATTRIBUTE NOTATION)
/*****
LIMIT 4
READ EMPLOYEES BY DEPT FROM 'T'
IF LEAVE-DUE GT 40
    DISPLAY NOTITLE 'EMPLOYEE' NAME          /* OVERRIDE STANDARD HEADER
        'LEAVE ACCUMULATED' LEAVE-DUE      /* OVERRIDE STANDARD HEADER
        '**'(10)(I)                        /* DISPLAY 10 '** INTENSIFIED
ELSE
    DISPLAY NAME LEAVE-DUE
/*****
END
    
```

EMPLOYEE	LEAVE ACCUMULATED	
-----		
LAVENDA	33	
BOYER	33	
CORREARD	45	*****
BOUVIER	19	

## Example 5

```

/* EXAMPLE 'DISEX5': DISPLAY (HORIZONTAL DISPLAY)
/*****
LIMIT 4
READ EMPLOYEES BY NAME
  DISPLAY NOTITLE NAME JOB-TITLE SALARY (1:2) 'CURR-CODE (1:2)
  SKIP 1
/*****
END
    
```

NAME	CURRENT POSITION	ANNUAL SALARY	CURRENCY CODE
ABELLAN	MAQUINISTA	1450000 1392000	PTA PTA
ACHIESON	DATA BASE ADMINISTRATOR	10500 11300	UKL UKL
ADAM	CHEF DE SERVICE	159980 0	FRA 0
ADKINSON	SALES PERSON	36000 33100	USD USD

## Example 6

```

/* EXAMPLE 'DISEX6': DISPLAY (VERTICAL AND HORIZONTAL DISPLAY)
/*****
LIMIT 1
READ EMPLOYEES BY NAME
  DISPLAY NOTITLE VERT AS CAPTIONED
    NAME CITY 'POSITION' JOB-TITLE
    HORIZ 'SALARY' SALARY (1:2) 'CURRENCY' CURR-CODE (1:2)
/*****
SKIP 1
END

```

NAME CITY POSITION	SALARY	CURRENCY
ABELLAN	1450000	PTA
MADRID	1392000	PTA
MAQUINISTA		

## Example 7

```

/* EXAMPLE 'DISEX7': DISPLAY (USING STATEMENT/ELEMENT PARAMETERS)
/*****
LIMIT 3
READ EMPLOYEES BY NAME
  DISPLAY NOTITLE (AL=16 GC=+ NL=8 SF=3 UC==)
    PERSONNEL-ID NAME TELEPHONE (LC=< TC=>)
/*****
END
    
```

PERSONNEL ID	NAME	+++++++TELEPHONE+++++++
		AREA CODE
=====	=====	=====
60008339	ABELLAN	<1 > <4356726 >
30000231	ACHIESON	<0332 > <523341 >
50005800	ADAM	<1033 > <44864858 >